

COMP4075: Lecture 7

Functional Programming Patterns: Functor, Foldable, and Friends

Henrik Nilsson

University of Nottingham, UK

COMP4075: Lecture 7 – p.1/40

Type Classes and Patterns

- In Haskell, many functional programming patterns are captured through specific type classes.
- Additionally, the type class mechanism itself and the fact that overloading is prevalent in Haskell give rise to other programming patterns.

COMP4075: Lecture 7 – p.2/40

Semigroups and Monoids (1)

Semigroups and monoids are algebraic structures:

- **Semigroup**: a set (type) S with an **associative** binary operation $\cdot : S \times S \rightarrow S$:

$$\forall a, b, c \in S : (a \cdot b) \cdot c = a \cdot (b \cdot c)$$

- **Monoid**: a semigroup with an **identity element**:

$$\exists e \in S, \forall a \in S : e \cdot a = a \cdot e = a$$

COMP4075: Lecture 7 – p.3/40

Semigroups and Monoids (2)

- Semigroups and monoids are patterns that appear frequently in everyday programming.
- Being explicit about when such structures are used
 - makes code clearer
 - offer opportunities for reuse
- The standard Haskell libraries provide type classes to capture these notions.

COMP4075: Lecture 7 – p.4/40

Class *Semigroup*

Class definition (most important methods):

```
class Semigroup a where
  (◇)      :: a → a → a
  sconcat :: NonEmpty a → a
```

Minimum complete definition: (◇) (ASCII: <>)
(There is thus a default definition for *sconcat*.)

NonEmpty is the non-empty list type:

```
data NonEmpty a = a :| [a]
```

COMP4075: Lecture 7 – p.5/40

Instances of *Semigroup* (2)

Addition and multiplication are associative; a numeric type with either operation forms a semigroup.

But which one to pick? Both are equally useful!

Idea:

- *Sum* a: the semigroup (a, (+))
- *Product* a: the semigroup (a, (*))

COMP4075: Lecture 7 – p.7/40

Instances of *Semigroup* (1)

A list [a] is a semigroup (for any type a):

```
instance Semigroup [a] where
  (◇) = (++)
```

Maybe a is a semigroup if a is one:

```
instance Semigroup a
  ⇒ Semigroup (Maybe a) where
  Nothing ◇ y      = y
  x      ◇ Nothing = x
  Just x  ◇ Just y  = x ◇ y
```

COMP4075: Lecture 7 – p.6/40

Instances of *Semigroup* (3)

Semigroup instances for *Sum* a and *Product* a:

```
instance Num a ⇒ Semigroup (Sum a) where
  (◇) = (+)
instance Num a ⇒ Semigroup (Product a) where
  (◇) = (*)
```

COMP4075: Lecture 7 – p.8/40

Instances of *Semigroup* (4)

Similarly, any type with a total ordering forms a semigroup with maximum or minimum as the associative operation:

- *Max a*: the semigroup (a, max)
- *Min a*: the semigroup (a, min)

Semigroup instances:

```
instance Ord a => Semigroup (Max a) where
  (◇) = max
```

```
instance Ord a => Semigroup (Min a) where
  (◇) = min
```

COMP4075: Lecture 7 – p.9/40

Exercise: *Semigroup* Instances

What is the value of the following expressions?

```
[1, 3, 7] ◇ [2, 4]
Sum 3 ◇ Sum 1 ◇ Sum 5
Just (Max 42) ◇ Nothing ◇ Just (Max 3)
sconcat (Product 2 :) [Product 3, Product 4]
([1], Product 2) ◇ ([2, 3], Product 3)
((1:) ◇ tail) [4, 5, 6]
```

COMP4075: Lecture 7 – p.11/40

Instances of *Semigroup* (5)

All products of semigroups are semigroups; e.g.:

```
instance (Semigroup a, Semigroup b)
  => Semigroup (a, b) where
  (x, y) ◇ (x', y') = (x ◇ x', y ◇ y')
```

$a \rightarrow b$ is a semigroup if the range b is a semigroup:

```
instance Semigroup b
  => Semigroup (a → b) where
  f ◇ g = λx → f x ◇ g x
```

COMP4075: Lecture 7 – p.10/40

Class *Monoid*

Recall: A monoid is a semigroup with an identity element:

```
class Semigroup a => Monoid a where
  mempty :: a
  mappend :: a → a → a
  mappend = (◇)
  mconcat :: [a] → a
  mconcat = foldr mappend mempty
```

Minimum complete definition: *mempty*

COMP4075: Lecture 7 – p.12/40

Instances of *Monoid* (1)

A list $[a]$ is the archetypical example of a monoid:

```
instance Monoid [a] where
  empty = []
```

Any semigroup can be turned into a monoid by adjoining an identity element:

```
instance Semigroup a
  => Monoid (Maybe a) where
  empty = Nothing
```

COMP4075: Lecture 7 – p.13/40

Instances of *Monoid* (3)

Monoid instances for *Min a* and *Max a*:

```
instance (Ord a, Bounded a) =>
  Monoid (Min a) where
  empty = maxBound

instance (Ord a, Bounded a) =>
  Monoid (Max a) where
  empty = minBound
```

COMP4075: Lecture 7 – p.15/40

Instances of *Monoid* (2)

Monoid instances for *Sum a* and *Product a*:

```
instance Num a => Monoid (Sum a) where
  empty = Sum 0

instance Num a => Monoid (Product a) where
  empty = Product 1
```

COMP4075: Lecture 7 – p.14/40

Instances of *Monoid* (4)

All products of monoids are monoids; e.g.:

```
instance (Monoid a, Monoid b)
  => Monoid (a, b) where
  empty = (empty, empty)
```

$a \rightarrow b$ is a monoid if the range b is a monoid:

```
instance Monoid b => Monoid (a -> b) where
  empty _ = empty
```

COMP4075: Lecture 7 – p.16/40

Functors (1)

A Functor is a notion that originated in a branch of mathematics called Category Theory.

However, for our purposes, we can think of functors as type constructors T (of arity 1) for which a function map can be defined:

$$map :: (a \rightarrow b) \rightarrow Ta \rightarrow Tb$$

that satisfies the following laws:

$$\begin{aligned} map\ id &= id \\ map(f \circ g) &= map\ f \circ map\ g \end{aligned}$$

COMP4075: Lecture 7 – p.17/40

Functors (3)

And trees; e.g.:

```
data Tree a = Leaf a
            | Node (Tree a) a (Tree a)
mapTree :: (a -> b) -> Tree a -> Tree b
mapTree f (Leaf x)      = Leaf (f x)
mapTree f (Node l x r) = Node (mapTree f l)
                              (f x)
                              (mapTree f r)
```

COMP4075: Lecture 7 – p.19/40

Functors (2)

Common examples of functors include (but are not limited to) **container types** like lists:

```
mapList :: (a -> b) -> [a] -> [b]
mapList _ []      = []
mapList f (x : xs) = f x : mapList f xs
```

COMP4075: Lecture 7 – p.18/40

Class Functor (1)

Of course, the notion of a functor is captured by a type class in Haskell:

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
  (<$) :: a -> f b -> f a
  (<$) = fmap o const
```

COMP4075: Lecture 7 – p.20/40

Class *Functor* (2)

There is also an infix version that can be viewed as function application lifted over a functor:

$$\begin{aligned}(\langle \$ \rangle) &:: (a \rightarrow b) \rightarrow f\ a \rightarrow f\ b \\ \langle \$ \rangle &= fmap\end{aligned}$$

Compare the standard infix function application operator:

$$(\$) :: (a \rightarrow b) \rightarrow a \rightarrow b$$

COMP4075: Lecture 7 – p.21/40

Instances of *Functor* (1)

As noted, list is a functor:

```
instance Functor [] where
    fmap = listMap
```

Maybe is also a functor:

```
instance Functor Maybe where
    fmap _ Nothing = Nothing
    fmap f (Just x) = Just (f x)
```

COMP4075: Lecture 7 – p.23/40

Class *Functor* (3)

However, Haskell's type system is not powerful enough to enforce the functor laws.

In general, the programmer is responsible for ensuring that an instance respects all laws associated with a type class.

Note that the type of *fmap* can be read:

$$(a \rightarrow b) \rightarrow (f\ a \rightarrow f\ b)$$

That is, we can see *fmap* as promoting a function to work in a different context.

COMP4075: Lecture 7 – p.22/40

Instances of *Functor* (2)

Container types are in general instances of functor, including *Array*:

```
instance Functor (Array i) where ...
```

E.g, given a matrix $m :: \text{Array } (\text{Int}, \text{Int}) \text{ Double}$, we can double all elements:

$$fmap (*2) m$$

COMP4075: Lecture 7 – p.24/40

Instances of *Functor* (3)

As functors are so common, there is a GHC extension for deriving *Functor* instances in standard cases.

For example, the functor instance for our tree type can be derived:

```
data Tree a = Leaf a
            | Node (Tree a) a (Tree a)
  deriving Functor
```

COMP4075: Lecture 7 – p.25/40

Nesting functors (1)

In practice, functors often appear nested inside other functors, e.g.

```
mxs :: [Maybe Double]
```

Such a structure can of course be processed by repeated mapping, e.g.:

```
fmap (fmap (*2)) mxs
```

One reading of this is “use *fmap* to lift *(*2)* to work on *Maybe*, and then map that over the list”.

COMP4075: Lecture 7 – p.27/40

Instances of *Functor* (4)

The type of functions from a given domain is an example of a functor that is **not a container** type. Map is just function composition:

```
instance Functor ((->) a) where
  fmap = (.)
```

Note that a **curried** function type, like

$$a \rightarrow b \rightarrow c = a \rightarrow (b \rightarrow c)$$

thus is a **nesting** or **composition** of functors:

$$(((\rightarrow) a) (((\rightarrow) b) c)) = (((\rightarrow) a) \circ (((\rightarrow) b))) c$$

COMP4075: Lecture 7 – p.26/40

Nesting functors (2)

However, in general $f (g a) = (f \circ g) a$, meaning

$$fmap (fmap (*2)) = (fmap \circ fmap) (*2)$$

suggesting the following combinator:

```
(<$$>) :: (Functor f, Functor g) =>
         (a -> b) -> f (g a) -> f (g b)
(<$$>) = fmap \circ fmap
```

This allows us to simplify $fmap (fmap (*2)) mxs$ to

```
(*2) <$$> mxs
```

COMP4075: Lecture 7 – p.28/40

Nesting functors (3)

Note that the composition of *fmaps* is mirrored by composition of functors at the type level:

$$[Maybe\ a] = []\ (Maybe\ a) = ([\ \circ\ Maybe)\ a$$

This can be generalized to any number levels; e.g.

$$\begin{aligned}(\langle\ \$\ \$\ \$\ \rangle) &= fmap\ \circ\ fmap\ \circ\ fmap \\ (*2)\ \langle\ \$\ \$\ \$\ \rangle\ [[1,2], [3]], [[4]], [[5]] \\ &\Rightarrow [[2,4], [6]], [[8]], [[10]]\end{aligned}$$

Data.Functor.Syntax defines $\langle\ \$\ \$\ \rangle$, $\langle\ \$\ \$\ \$\ \rangle$...

COMP4075: Lecture 7 – p.29/40

Class *Foldable* (1)

Class of data structures that can be folded to a summary value.

Many methods; minimal instance *foldMap*, *foldr*:

class *Foldable* *t* **where**

$$\begin{aligned}fold &:: Monoid\ m \Rightarrow t\ m \rightarrow m \\ foldMap &:: Monoid\ m \Rightarrow (a \rightarrow m) \rightarrow t\ a \rightarrow m \\ foldr &:: (a \rightarrow b \rightarrow b) \rightarrow b \rightarrow t\ a \rightarrow b \\ foldr' &:: (a \rightarrow b \rightarrow b) \rightarrow b \rightarrow t\ a \rightarrow b \\ foldl &:: (b \rightarrow a \rightarrow b) \rightarrow b \rightarrow t\ a \rightarrow b \\ foldl' &:: (b \rightarrow a \rightarrow b) \rightarrow b \rightarrow t\ a \rightarrow b\end{aligned}$$

COMP4075: Lecture 7 – p.31/40

Nesting functors (4)

Note that we also could have defined:

$$(\langle\ \$\ \$\ \rangle) = fmap\ fmap\ fmap$$

Why?

Exploiting that curried function types are composed functors, $\langle\ \$\ \$\ \rangle$, $\langle\ \$\ \$\ \$\ \rangle$... can compose functions where the second function has arity 2, 3, ...:

$$\begin{aligned}f &:: Bool \rightarrow Double \rightarrow Int \rightarrow Double \\ (>0)\ \langle\ \$\ \$\ \$\ \rangle\ f &:: Bool \rightarrow Double \rightarrow Int \rightarrow Bool\end{aligned}$$

This is often quite handy in practice.

COMP4075: Lecture 7 – p.30/40

Class *Foldable* (2)

(continued)

$$\begin{aligned}foldr1 &:: (a \rightarrow a \rightarrow a) \rightarrow t\ a \rightarrow a \\ foldl1 &:: (a \rightarrow a \rightarrow a) \rightarrow t\ a \rightarrow a \\ toList &:: t\ a \rightarrow [a] \\ null &:: t\ a \rightarrow Bool \\ length &:: t\ a \rightarrow Int \\ elem &:: Eq\ a \Rightarrow a \rightarrow t\ a \rightarrow Bool\end{aligned}$$

(Note that *length* should be understood as *size*.)

COMP4075: Lecture 7 – p.32/40

Class *Foldable* (3)

(continued)

```
maximum :: Ord a => t a -> a
minimum :: Ord a => t a -> a
sum      :: Num a => t a -> a
product  :: Num a => t a -> a
```

Note: *foldl* typically incurs a large space overhead due to laziness. The version with strict application of the operator, *foldl'* is typically preferable.

COMP4075: Lecture 7 – p.33/40

Instances of *Foldable* (2)

But there are also some instances that are less expected, e.g.:

- **instance** *Foldable* (*Either* *a*) **where** ...
- **instance** *Foldable* ((,) *a*) **where** ...

This has some arguably odd consequences:

```
length (1, 2)    => 1
sum (1, 2)       => 2
length (Left 1)  => 0
length (Right 2) => 1
```

COMP4075: Lecture 7 – p.35/40

Instances of *Foldable* (1)

All expected instances, e.g.:

- **instance** *Foldable* [] **where** ...
- **instance** *Foldable* *Maybe* **where** ...

And GHC extension allows deriving instances in many cases; e.g.

```
data Tree a = ... deriving Foldable
```

COMP4075: Lecture 7 – p.34/40

Example: Folding Over a Tree (1)

Consider:

```
data Tree a = Empty
            | Node (Tree a) a (Tree a)
            deriving (Show, Eq)
```

Let us make it an instance of *Foldable*:

```
instance Foldable Tree where
  foldMap f Empty = mempty
  foldMap f (Node l a r) =
    foldMap f l <math>\diamond</math> f a <math>\diamond</math> foldMap f r
```

COMP4075: Lecture 7 – p.36/40

Example: Folding Over a Tree (2)

We wish to compute the sum and max over a tree of *Int*. One way:

```
sumMax :: Tree Int → (Int, Int)
sumMax t = (foldl (+) 0 t, foldl max minBound t)
```

Another way, with a single traversal:

```
sumMax :: Tree Int → (Int, Int)
sumMax t = (sm, mx)
```

where

```
(Sum sm, Max mx) =
  foldMap (λn → (Sum n, Max n)) t
```

COMP4075: Lecture 7 – p.37/40

Aside: Foldable?

Note that the kind of “folding” captured by the class *Foldable* in general makes it impossible to recover the structure over which the “folding” takes place.

Such an operation is also known as “reduce” or “crush”, and some authors prefer to reserve the term “fold” for *catamorphisms*, where a separate combining function is given for each constructor, making it possible to recover the structure.

One might thus argue that *Reducible* or *Crushable* would have been a more precise name.

COMP4075: Lecture 7 – p.39/40

Example: Folding Over a Tree (3)

The latter can be generalized to e.g. computing the sum, product, min, and max in a single traversal:

```
foldMap
  (λn → (Sum n, Product n, Min n, Max n))
  t
```

COMP4075: Lecture 7 – p.38/40

MapReduce

Functional mapping and folding (reducing) inspired the MapReduce programming model; e.g.

- Google’s original MapReduce framework
- Apache Hadoop

Functional mapping and folding with *associative* operator (semigroup) is amenable to parallelization and distribution.

However, achieving scalability in practice required both careful engineering of the frameworks as such, and a good understanding of how to use them on part of the user.

COMP4075: Lecture 7 – p.40/40